
ExternalService
An Elixir library for safely using external service or API using customizable retry logic, automatic rate limiting, and the circuit breaker pattern. Calls to external services can be synchronous, asynchronous background tasks, or multiple calls can be made in parallel for MapReduce style processing.
Overview
ExternalService is, in essence, the combination of two techniques: retrying individual requests that have failed and preventing cascading failures with circuit breakers. The basic approach to using ExternalService is to wrap all usages of any external services in an Elixir function and pass this function to the ExternalService.call function, together with some options that control the retry mechanism. By doing so, ExternalService can manage calls to the external API that you’re using, and apply retry logic to individual calls while ensuring that your application is protected from outages or other problems with the external API by using “circuit breakers” (described in more detail below).
In addition to the call function, ExternalService also provides a call_async function, which uses an Elixir Task to call the external service asynchronously, and a call_async_stream function, which makes multiple calls to the external service in parallel.
Both of these functions apply the same retry and “circuit breaker” mechanisms as the regular call function. Let’s look at these mechanisms in further detail.
Retrying failed requests
Many of the failures that occur when accessing external services are transient in nature. For example, there could be network congestion causing a request to timeout, or the service could be under heavy load and is not able to handle any more requests at the time. The best strategy for dealing with such transient failures is to simply retry the failed request – perhaps after a brief backoff period. The external_service package uses the retry library from Safwan Kamarrudin to automate retry logic. This retry library provides flexible configuration to control various aspects of retry logic, such as whether to use linear or exponential backoff, the maximum delay between retries, and the total amount of time to spend retrying before giving up.
Circuit breakers for preventing catastrophic cascade
The Circuit Breaker pattern was first described in Michael Nygard's landmark book Release It! and was later popularized by Martin Fowler on his bliki post. To quote Nygard, “Circuit breakers are a way to automatically degrade functionality when the system is under stress.”
The Circuit Breaker pattern is modeled after electrical circuit breakers, which are designed to protect electrical circuits from damage caused by excess current. Like these electrical circuit breakers, software “circuit breakers” are designed to protect the system at large from damage caused by faults in a component of the system. By protecting calls to external services, the circuit breaker can monitor for failures. If the failures reach some given threshold, the circuit breaker “trips” and further calls to the service will immediately return an error without even making a call to the external API itself. After a configurable period of time, the circuit breaker is automatically reset and calls to the external service will once again be attempted with the same monitoring as before.
In contrast to retry logic, which is applied to each individual call to a service, the circuit breaker for a given service is global to the entire system. If a circuit trips, then it trips for all users of the associated service. This is a key feature of the Circuit Breaker pattern, and is what allows it to prevent cascading failures.
The external_service package uses the Erlang fuse library from Jesper Louis Andersen for implementing circuit breakers. To extend the electrical analogies, circuit breakers in the fuse library are called “fuses.” These fuses can be used to protect against cascading failure by first asking the fuse if it is OK to proceed using the :fuse.ask function. If this function returns :ok then it is OK to proceed to calling on the external service. If, on the other hand, it returns :blown, then the fuse has been tripped and it is not safe to call the external service. In this scenario, your code must have a fallback option to compensate for the fact that the external service is unavailable, which might mean returning cached data or indicating to the user that the functionality is not currently available.
What causes a fuse to trip?
When using a fuse, your application code must tell the fuse about any failures that occur. If you’ve asked the fuse if it is OK to proceed but then receive an error from the external service, your code should call the :fuse.melt function, which “melts the fuse a little bit”. Once the fuse has been “melted” enough times, the fuse is tripped and future calls to :fuse.ask for that fuse will return :blown.
ExternalService wraps the functionality provided by fuse in a convenient interface and automates the handling of the fuse so that you don’t need to explicitly call :fuse.ask or :fuse.melt in your code. Instead, you simply use the ExternalService.call function with the name of the fuse as the first argument, together with the function in which you’ve wrapped your call to the external API. Then, the ExternalService.call function will first ask the given fuse before making the call and will return {:error, :fuse_blown} if the fuse is blown. It will also automatically call :fuse.melt any time the call to the given function results in a retry. This eventually results in a blown fuse if there are enough failed requests to the service being protected by the fuse.
The only requirement for using a fuse for a particular service is that it must be initialized before using the service. This is done with the ExternalService.start/2 function, which takes the fuse name and options as arguments. The fuse name is an atom which must uniquely identify the external service to which the fuse applies. This function should be called in your application startup code, which is typically the Application.start function for your application.
Rate Limiting
Since version 0.8.0, ExternalService allows for rate limiting of calls to a service by specifying the rate limit as an optional argument to ExternalService.start. If the rate_limit option is passed to ExternalService.start, then all calls to the external service will be automatically rate-limited. Once the number of calls to the external service has exceeded the limits for a given time window, then ExternalService will delay the call to the service until the time window has expired and calls to the service are allowed again. The delay is accomplished using Elixir's Process.sleep/1 function, so if you are using ExternalService.call then the calling process is put to sleep for the specified time window. If, on the other hand, you are using ExternalService.call_async or ExternalService.call_async_stream, then it is the background process(es) that are put to sleep, and the calling process is not put to sleep in this case. In any case, the rate-limiting is applied to all calls to a particular service across your application so that you can rest assured that you will not violate the rate limits imposed by the external service that you are calling.
Applying rate limits to an external service is as simple as specifying the limits in ExternalService.start, as in the following example, which limits calls to :some_external_service a maximum of 10 times per second:
ExternalService.start(:some_external_service, rate_limit: {10, 1000})
Usage Examples
Below are some example usages of ExternalService that illustrate how to configure fuses and retries, and how to use the various forms of the call functions for using an external service reliably. Some of the examples are adapted from the Ropig code base that was the original use case for ExternalService, and they show how to use ExternalService for interacting with Google's Pub/Sub messaging service.
Fuse initialization
This example shows how to initialize the fuse for a service, as well as how to apply automatic rate limiting to that service.
defmodule PubSub do
  @fuse_name __MODULE__
  @fuse_options [
    # Tolerate 5 failures for every 1 second time window.
    fuse_strategy: {:standard, 5, 1_000},
    # Reset the fuse 5 seconds after it is blown.
    fuse_refresh: 5_000,
    # Limit to 100 calls per second.
    rate_limit: {100, 1_000}
  ]

  def start do
    ExternalService.start(@fuse_name, @fuse_options)
  end
end
Triggering a retry
This example illustrates the usage of the ExternalService.call function and some of the retry options that can be used to control retry behavior. Notice how we delegate to a named function in call.
defmodule PubSub do
  @retry_errors [
    408, # TIMEOUT
    429, # RESOURCE_EXHAUSTED
    499, # CANCELLED
    500, # INTERNAL
    503, # UNAVAILABLE
    504, # DEADLINE_EXCEEDED
  ]
  @retry_opts %ExternalService.RetryOptions{
    # Use linear backoff. Exponential backoff is also available.
    backoff: {:linear, 100, 1},
    # Stop retrying after 5 seconds.
    expiry: 5_000,
  }

  def publish(message, topic) do
    ExternalService.call(PubSub, @retry_opts, fn -> try_publish(message, topic) end)
  end

  defp try_publish(message, topic) do
    message
    |> Kane.Message.publish(%Kane.Topic{name: topic})
    |> case do
      {:error, reason, code} when code in @retry_errors ->
        {:retry, reason}
      kane_result ->
        # If not a retriable error, just return the result.
        kane_result
    end
  end
end
We wrap our call to the external service in a function and pass this function to ExternalService.call with retry options as the second argument. Importantly, we use a special return value from our anonymous function to trigger a retry. A retry is triggered by one of three mechanisms:
	the function returns :retry
	the function returns a tuple of the form {:retry, reason}
	the function raises an Elixir RuntimeError, or another exception type specified in the rescue_only option

In our example code, we examine the result of calling Kane.Message.publish and if it is an error response with an error code that matches one of our predetermined @retry_errors, we then trigger a retry.
Not all failed requests should be retried, of course. Some failures are due to bugs in the calling code; such calls can never succeed and therefore should not be retried. In our case, we consulted the documentation for Google Pub/Sub to determine which error codes should result in a retry. You will have to decide on a strategy to determine what error conditions are retriable for your service.
Error handling
Although the retry mechanism goes a long way towards eliminating transient failures, there will be times when a service is unavailable for a long enough time that retries ultimately fail. If, for example, a request has been retried several times and the time spent on retries exceeds the configured expiry time for that request, then ExternalService.call will give up and return the tuple {:error, :retries_exhausted}.
Another failure scenario is when there are many different processes using an external service concurrently, such as for example many different web requests. If the service or API being used is temporarily unable to handle requests, then all of these concurrent calls to the service will eventually trip the circuit breaker associated with that external service. Then, ExternalService.call will return the tuple {:error, :fuse_blown}. Further calls to the service at this point would immediately result in the {:error, :fuse_blown} until the fuse is reset, which happens automatically after the configured fuse_refresh time for the fuse.
It is up to the caller to determine how to handle these errors. A web application might, for example, log the error and return a 503 Service Unavailable status code. Background jobs could log the error and pause until the service is fully functioning again. This example shows a Phoenix controller that uses the PubSub.publish function from above and handles failed requests by returning a 503 status code.
defmodule MyApp.MyController do
  use MyAppWeb, :controller
  require Logger

  @topic "some_topic"

  def create(conn, %{"message" => message}) do
    case PubSub.publish(message, @topic) do
      {:ok, _result} ->
        send_resp(conn, 201, "")

      {:error, {:retries_exhausted, reason}} ->
        Logger.error("Retries exhausted while trying to publish to #{@topic}: #{inspect(reason)}")
        send_resp(conn, 503, "")

      {:error, {:fuse_blown, fuse_name}} ->
        Logger.error("Fuse blown while trying to publish to #{@topic}: #{inspect(fuse_name)}")
        send_resp(conn, 503, "")

      error ->
        # If we got here it means that we did not an :ok response from Kane, nor did we get one of
        # the error tuples meaning retries_exhausted or fuse_blown, so it must be some other kind
        # of non-retriable error response from Kane itself. Log the error and send back a 503.
        Logger.error("Unknown error while trying to publish to #{@topic}: #{inspect(error)}")
        send_resp(conn, 503, "")
    end
  end
end
Cleaning up error handling with ExternalService.call!
As seen in the above example, error handling code can be somewhat intricate because we must distinguish between error responses that are created by ExternalService.call versus responses that originate from the service itself. In cases like this it can be useful to use ExternalService.call! so that error responses created by ExternalService are raised as exceptions instead of returned as error tuples. To show how this works, let's add a new publish! function to the PubSub module, which is just like publish except that it uses call! instead of call.
defmodule PubSub do
  # same @retry_errors and @retry_opts from above example...

  # Will raise a `RetriesExhaustedError` or `FuseBlownError` in event of failure.
  def publish!(message, topic) do
    ExternalService.call!(PubSub, @retry_opts, fn -> try_publish(message, topic) end)
  end
end
Now let's see how this impacts our calling code in the hypothetical controller:
defmodule MyApp.MyController do
  use MyAppWeb, :controller
  require Logger

  @topic "some_topic"

  def create(conn, %{"message" => message}) do
    try do
      case PubSub.publish!(message, @topic) do
        {:ok, _result} ->
          send_resp(conn, 201, "")

        error ->
          Logger.error("Unknown error while trying to publish to #{@topic}: #{inspect(error)}")
          send_resp(conn, 503, "")
      end
    rescue
      e in [ExternalService.RetriesExhaustedError, ExternalService.FuseBlownError] ->
        Logger.error(Exception.format(:error, e))
        send_resp(conn, 503, "")
    end
  end
end
By using call!, it is much more apparent which kinds of errors are coming from the actual service being used, rather than those that are created by ExternalService.
Asynchronous calls
In addition to the call function demonstrated above, the ExternalService module also provides call_async and call_async_stream:
	call_async - asynchronous version of call that returns an Elixir Task that can be used to retrieve the result
	call_async_stream - parallel, streaming version of call that is modeled after Elixir's built-in Task.async_stream function

Both of these asynchronous functions apply the same retry and circuit breaker mechanisms as the synchronous call function, so any necessary retries are performed transparently in the background task(s).
In the code examples that follow, these asynchronous forms of call are illustrated using the same @retry_errors, @retry_opts, and try_publish/2 function from the previous example above.
defmodule PubSub do
  # same @retry_errors and @retry_opts from above example...

  # Returns an Elixir `Task`, which can be used to retrieve the result,
  # using `Task.await`, for example.
  def publish_async(message, topic) do
    ExternalService.call_async(PubSub, @retry_opts, fn -> try_publish(message, topic) end)
  end

  # Publish many messages in parallel and return a Stream of results as described by `Task.async_stream`.
  def publish_async_stream(messages, topic) when is_list(messages) do
    ExternalService.call_async_stream(messages, PubSub, @retry_opts, fn message ->
      try_publish(message, topic)
    end)
  end
end
Using the async version is a simple means of achieving paralellism since other work can be accomplished while the external calls are taking place. For example:
task = PubSub.publish_async("Hello", "world")
do_other_things()
case Task.await(task) do
  {:ok, _result} ->
    :ok
  {:error, {:retries_exhausted, reason}} ->
    :error
  {:error, {:fuse_blown, fuse_name}} ->
    :error
end
Documentation
See my blog post for overview documentation, and then check out the API reference for full details.
Installation
external_service is available in Hex, and can be installed
by adding external_service to your list of dependencies in mix.exs:
def deps do
  [{:external_service, "~> 1.0.1"}]
end
Documentation can be generated with ExDoc
and published on HexDocs. Once published, the docs can
be found at https://hexdocs.pm/external_service.
Sponsored by Ropig http://ropig.com

    error()
  

    Union type representing all the possible error tuple return values





  
    fuse_blown()
  

    Error tuple returned when a fuse has been melted enough times that the fuse is blown





  
    fuse_name()
  

    Name of a fuse





  
    fuse_not_found()
  

    Error tuple returned when a fuse has not been initialized with ExternalService.start/1





  
    fuse_strategy()
  

    Strategy controlling fuse behavior.





  
    options()
  

    Options used for controlling circuit breaker and rate-limiting behavior.





  
    rate_limit()
  

    A tuple specifying rate-limiting behavior.





  
    retriable_function()
  




  
    retriable_function_result()
  




  
    retries_exhausted()
  

    Error tuple returned when the allowable number of retries has been exceeded

    call(fuse_name, retry_opts \\ %RetryOptions{}, function)
  

    Given a fuse name and retry options execute a function handling any retry and circuit breaker
logic.





  
    call!(fuse_name, retry_opts \\ %RetryOptions{}, function)
  

    Like call/3, but raises an exception if retries are exhausted or the fuse is blown.





  
    call_async(fuse_name, retry_opts \\ %RetryOptions{}, function)
  

    Asynchronous version of ExternalService.call.





  
    call_async_stream(enumerable, fuse_name, function)
  

    Parallel, streaming version of ExternalService.call.





  
    call_async_stream(enumerable, fuse_name, retry_opts_or_async_opts, function)
  

    Parallel, streaming version of ExternalService.call.





  
    call_async_stream(enumerable, fuse_name, retry_opts, async_opts, function)
  

    Parallel, streaming version of ExternalService.call.





  
    reset_fuse(fuse_name)
  

    Resets the given fuse.





  
    start(fuse_name, options \\ [])
  

    Initializes a new fuse for a specific service.





  
    stop(fuse_name)
  

    Stops the fuse for a specific service.

      Link to this type

      Specs

      
          error() :: retries_exhausted() | fuse_blown() | fuse_not_found()

      


Union type representing all the possible error tuple return values
  


  
    
      
      Link to this type
    
    fuse_blown()

      
       
       View Source
     
  


  
      Specs

      
          fuse_blown() :: {:error, {:fuse_blown, fuse_name()}}

      


Error tuple returned when a fuse has been melted enough times that the fuse is blown
  


  
    
      
      Link to this type
    
    fuse_name()

      
       
       View Source
     
  


  
      Specs

      
          fuse_name() :: term()

      


Name of a fuse
  


  
    
      
      Link to this type
    
    fuse_not_found()

      
       
       View Source
     
  


  
      Specs

      
          fuse_not_found() :: {:error, {:fuse_not_found, fuse_name()}}

      


Error tuple returned when a fuse has not been initialized with ExternalService.start/1
  


  
    
      
      Link to this type
    
    fuse_strategy()

      
       
       View Source
     
  


  
      Specs

      
          fuse_strategy() ::
  {:standard, max_melt_attempts :: pos_integer(), time_window :: pos_integer()}
  | {:fault_injection, rate :: float(), max_melt_attempts :: pos_integer(),
     time_window :: pos_integer()}

      


Strategy controlling fuse behavior.
  


  
    
      
      Link to this type
    
    options()

      
       
       View Source
     
  


  
      Specs

      
          options() :: [
  fuse_strategy: fuse_strategy(),
  fuse_refresh: pos_integer(),
  rate_limit: rate_limit()
]

      


Options used for controlling circuit breaker and rate-limiting behavior.
See the fuse docs for further information about available fuse options.
  


  
    
      
      Link to this type
    
    rate_limit()

      
       
       View Source
     
  


  
      Specs

      
          rate_limit() :: {limit :: pos_integer(), time_window :: pos_integer()}

      


A tuple specifying rate-limiting behavior.
The first element of the tuple is the number of calls to allow in a given time window.
The second element is the time window in milliseconds.
  


  
    
      
      Link to this type
    
    retriable_function()

      
       
       View Source
     
  


  
      Specs

      
          retriable_function() :: (() -> retriable_function_result())

      


  


  
    
      
      Link to this type
    
    retriable_function_result()

      
       
       View Source
     
  


  
      Specs

      
          retriable_function_result() ::
  :retry | {:retry, reason :: any()} | (function_result :: any())

      


  


  
    
      
      Link to this type
    
    retries_exhausted()

      
       
       View Source
     
  


  
      Specs

      
          retries_exhausted() :: {:error, {:retries_exhausted, reason :: any()}}

      


Error tuple returned when the allowable number of retries has been exceeded

      Link to this function
    
    call(fuse_name, retry_opts \\ %RetryOptions{}, function)

      
       
       View Source
     
  


  
      Specs

      
          call(fuse_name(), ExternalService.RetryOptions.t(), retriable_function()) ::
  error() | (function_result :: any())

      


Given a fuse name and retry options execute a function handling any retry and circuit breaker
logic.
ExternalService.start must be run with the fuse name before using call.
The provided function can indicate that a retry should be performed by returning the atom
:retry or a tuple of the form {:retry, reason}, where reason is any arbitrary term, or by
raising a RuntimeError. Any other result is considered successful so the operation will not be
retried and the result of the function will be returned as the result of call.
  


    
  
    
      
      Link to this function
    
    call!(fuse_name, retry_opts \\ %RetryOptions{}, function)

      
       
       View Source
     
  


  
      Specs

      
          call!(fuse_name(), ExternalService.RetryOptions.t(), retriable_function()) ::
  function_result :: any() | no_return()

      


Like call/3, but raises an exception if retries are exhausted or the fuse is blown.
  


    
  
    
      
      Link to this function
    
    call_async(fuse_name, retry_opts \\ %RetryOptions{}, function)

      
       
       View Source
     
  


  
      Specs

      
          call_async(fuse_name(), ExternalService.RetryOptions.t(), retriable_function()) ::
  Task.t()

      


Asynchronous version of ExternalService.call.
Returns a Task that may be used to retrieve the result of the async call.

    call_async_stream(enumerable, fuse_name, function)

      
       
       View Source
     
  


  
      Specs

      
          call_async_stream(
  Enumerable.t(),
  fuse_name(),
  (any() -> retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of ExternalService.call.
See call_async_stream/5 for full documentation.
  


  
    
      
      Link to this function
    
    call_async_stream(enumerable, fuse_name, retry_opts_or_async_opts, function)

      
       
       View Source
     
  


  
      Specs

      
          call_async_stream(
  Enumerable.t(),
  fuse_name(),
  ExternalService.RetryOptions.t() | (async_opts :: list()),
  (any() -> retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of ExternalService.call.
See call_async_stream/5 for full documentation.
  


  
    
      
      Link to this function
    
    call_async_stream(enumerable, fuse_name, retry_opts, async_opts, function)

      
       
       View Source
     
  


  
      Specs

      
          call_async_stream(
  Enumerable.t(),
  fuse_name(),
  ExternalService.RetryOptions.t(),
  async_opts :: list(),
  (any() -> retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of ExternalService.call.
This function uses Elixir's built-in Task.async_stream/3 function and the description below is
taken from there.
Returns a stream that runs the given function function concurrently on each
item in enumerable.
Each enumerable item is passed as argument to the given function function
and processed by its own task. The tasks will be linked to the current
process, similarly to async/1.

    reset_fuse(fuse_name)

      
       
       View Source
     
  


  
      Specs

      
          reset_fuse(fuse_name()) :: :ok

      


Resets the given fuse.
After reset, the fuse will be unbroken with no melts.
  


    
  
    
      
      Link to this function
    
    start(fuse_name, options \\ [])

      
       
       View Source
     
  


  
      Specs

      
          start(fuse_name(), options()) :: :ok

      


Initializes a new fuse for a specific service.
The fuse_name is a term that uniquely identifies an external service within the scope of
an application.
The options argument allows for controlling the circuit breaker behavior and rate-limiting
behavior when making calls to the external service. See options/0 for details.
  


  
    
      
      Link to this function
    
    stop(fuse_name)

      
       
       View Source
     
  


  
      Specs

      
          stop(fuse_name()) :: :ok

      


Stops the fuse for a specific service.

Defines a gateway to an external service.
ExternalService.Gateway allows for defining module-based gateways to external services.
Instead of explicitly starting a fuse with its configuration and separately passing in retry
options on each call to the service, a module-based gateway allows one to specify default fuse
and retry options at the module level.
When a module uses the ExternalService.Gateway module, an implementation of the
ExternalService.Gateway behaviour will be generated using the fuse, retry, and rate-limit
options provided to the use ExternalService.Gateway statement. See the documentation for the
various callbacks in this module for more details.
Example
defmodule MyApp.SomeService do
  use ExternalService.Gateway,
    fuse: [
      # Tolerate 5 failures for every 1 second time window.
      strategy: {:standard, 5, 10_000},
      # Reset the fuse 5 seconds after it is blown.
      refresh: 5_000
    ],
    # Limit to 5 calls per second.
    rate_limit: {5, :timer.seconds(1)},
    retry: [
      # Use linear backoff. Exponential backoff is also available.
      backoff: {:linear, 100, 1},
      # Stop retrying after 5 seconds.
      expiry: 5_000
    ]

  def call_the_service(params) do
    external_call fn ->
      # Call the service with params, then return the result or :retry.
      case do_call(params) do
        {:ok, result} -> {:ok, result}
        {:error, reason} -> {:retry, reason}
      end
    end
  end
end
Initialization and configuration
Gateways must be started (preferably under a supervisor) before being used.
To initialize a gateway with its default configuration, just add the gateway module to the
top-level supervisor for your application:
children = [
  MyApp.SomeService
]

Supervisor.start_link(children, strategy: :one_for_one)
It is also possible to override the default configuration for the gateway by passing options to
the child specification that is passed to the supervisor. This can be useful for using different
configuration in the test environment. For example:
some_service_config = Application.get_env(:my_app, :some_service, [])

children = [
  {MyApp.SomeService, some_service_config}
]

Supervisor.start_link(children, strategy: :one_for_one)

      Callbacks
    


  
    external_call(retriable_function)
  

    Invoked to call the given function using the retry options configured for the gateway.





  
    external_call(t, retriable_function)
  

    Invoked to call the given function using custom retry options.





  
    external_call!(retriable_function)
  

    Like external_call/1, but raises an exception if retries are exhausted or the fuse is blown.





  
    external_call!(t, retriable_function)
  

    Like external_call/2, but raises an exception if retries are exhausted or the fuse is blown.





  
    external_call_async(retriable_function)
  

    Asynchronous version of external_call/1.





  
    external_call_async(t, retriable_function)
  

    Asynchronous version of external_call/2.





  
    external_call_async_stream(t, function)
  

    Parallel, streaming version of external_call/1.





  
    external_call_async_stream(t, arg2, function)
  

    Parallel, streaming version of external_call/2.





  
    external_call_async_stream(t, t, async_opts, function)
  

    Parallel, streaming version of external_call/2.

      Link to this callback
    
    external_call(retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call(ExternalService.retriable_function()) ::
  ExternalService.error() | (function_result :: any())

      


Invoked to call the given function using the retry options configured for the gateway.
See ExternalService.call/3 for more information.
  


  
    
      
      Link to this callback
    
    external_call(t, retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call(
  ExternalService.RetryOptions.t(),
  ExternalService.retriable_function()
) :: ExternalService.error() | (function_result :: any())

      


Invoked to call the given function using custom retry options.
See ExternalService.call/3 for more information.
  


  
    
      
      Link to this callback
    
    external_call!(retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call!(ExternalService.retriable_function()) ::
  function_result :: any() | no_return()

      


Like external_call/1, but raises an exception if retries are exhausted or the fuse is blown.
See ExternalService.call!/3 for more information.
  


  
    
      
      Link to this callback
    
    external_call!(t, retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call!(
  ExternalService.RetryOptions.t(),
  ExternalService.retriable_function()
) :: function_result :: any() | no_return()

      


Like external_call/2, but raises an exception if retries are exhausted or the fuse is blown.
See ExternalService.call!/3 for more information.
  


  
    
      
      Link to this callback
    
    external_call_async(retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call_async(ExternalService.retriable_function()) :: Task.t()

      


Asynchronous version of external_call/1.
Returns a Task that may be used to retrieve the result of the async call.
See ExternalService.call_async for more information.
  


  
    
      
      Link to this callback
    
    external_call_async(t, retriable_function)

      
       
       View Source
     
  


  
      Specs

      
          external_call_async(
  ExternalService.RetryOptions.t(),
  ExternalService.retriable_function()
) :: Task.t()

      


Asynchronous version of external_call/2.
Returns a Task that may be used to retrieve the result of the async call.
See ExternalService.call_async for more information.
  


  
    
      
      Link to this callback
    
    external_call_async_stream(t, function)

      
       
       View Source
     
  


  
      Specs

      
          external_call_async_stream(
  Enumerable.t(),
  (any() -> ExternalService.retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of external_call/1.
See ExternalService.call_async_stream/5 for more information.
  


  
    
      
      Link to this callback
    
    external_call_async_stream(t, arg2, function)

      
       
       View Source
     
  


  
      Specs

      
          external_call_async_stream(
  Enumerable.t(),
  ExternalService.RetryOptions.t() | (async_opts :: list()),
  (any() -> ExternalService.retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of external_call/2.
See ExternalService.call_async_stream/5 for more information.
  


  
    
      
      Link to this callback
    
    external_call_async_stream(t, t, async_opts, function)

      
       
       View Source
     
  


  
      Specs

      
          external_call_async_stream(
  Enumerable.t(),
  ExternalService.RetryOptions.t(),
  async_opts :: list(),
  (any() -> ExternalService.retriable_function_result())
) :: Enumerable.t()

      


Parallel, streaming version of external_call/2.
See ExternalService.call_async_stream/5 for more information.
  

        

      
  
    
ExternalService.RetryOptions    



      
Options used for controlling retry logic.
See the retry docs for information about the available
retry options.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    backoff()
  

    A tuple describing the backoff strategy for increasing delay between retries.





  
    t()
  

    Struct representing the retry options to apply to calls to external services.




  

  
    
      Functions
    


  
    new(opts)
  



  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    backoff()

      
       
       View Source
     
  


  
      Specs

      
          backoff() ::
  {:exponential, initial_delay :: pos_integer()}
  | {:linear, initial_delay :: pos_integer(), factor :: pos_integer()}

      


A tuple describing the backoff strategy for increasing delay between retries.
The first element of the tuple must be one of the atoms :exponential or :linear.
In both cases, the second element of the tuple is an integer representing the initial delay
between retries, in milliseconds.
For linear delay, there is also a third element in the tuple, which is a number representing
the factor that the initial delay will be multiplied by on each successive retry.
  


  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %ExternalService.RetryOptions{
  backoff: backoff(),
  cap: pos_integer() | nil,
  expiry: pos_integer() | nil,
  randomize: boolean(),
  rescue_only: [module()]
}

      


Struct representing the retry options to apply to calls to external services.
	backoff: tuple describing the backoff strategy (see backoff/0)
	randomize: boolean indicating whether or not delays between retries should be randomized
	expiry: limit total length of time to allow for retries to the specified time budget
	  milliseconds
	cap: limit maximum amount of time between retries to the specified number of milliseconds
	rescue_only: retry only on exceptions matching one of the list of provided exception types,
  (defaults to [RuntimeError])

  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    new(opts)

      
       
       View Source
     
  


  

  

        

      
  
    
ExternalService.FuseBlownError exception    



      
Exception raised by ExternalService.call!/3 when a fuse has been melted enough times that
the fuse is blown.
      


  
    
ExternalService.FuseNotFoundError exception    



      
Exception raised by ExternalService.call!/3 when a fuse has not been initialized with
ExternalService.start/1.
      


  
    
ExternalService.RetriesExhaustedError exception    



      
Exception raised by ExternalService.call!/3 when the allowable number of retries has been
exceeded.
      


  
    
ExternalService.FuseBlownError exception    



      
Exception raised by ExternalService.call!/3 when a fuse has been melted enough times that
the fuse is blown.
      


  
    
ExternalService.FuseNotFoundError exception    



      
Exception raised by ExternalService.call!/3 when a fuse has not been initialized with
ExternalService.start/1.
      


  
    
ExternalService.RetriesExhaustedError exception    



      
Exception raised by ExternalService.call!/3 when the allowable number of retries has been
exceeded.

